

Čas br.4

Pojmovi:

- Sistemske varijable
- Booleovi izrazi / Kondicionali

Sistemske varijable

Kao što smo vidjeli s mouseX i mouseY, Processing posjeduje neke ugrađene varijable koje su nam dostupne za koristiti. Obično su to neki podaci vezani za sam program, na primjer širina i visina prozora. Kada imenujemo svoje varijable bitno je da izbjegnemo davanje imena sistemskih varijabli.

Lista sistemski varijabli koje ćemo najčešće koristiti:

width, height, frameCount, frameRate, screen.width, screen.height, key, keyCode, keyPressed, mousePressed, mouseButton.

Primjer:

```
void setup() {  
  size(200,200);  
  frameRate(30);  
}  
void draw() {  
  background(100);  
  stroke(255);  
  fill(frameCount/2);  
  rectMode(CENTER);  
  rect(width/2,height/2,mouseX + 10,mouseY + 10);  
}  
void keyPressed() {  
  println(key);  
}
```

Booleovi izrazi

Booleov izraz (nazvani prema matematičaru George Boole-u) je izrazi koji se evaluira kao ili tačan ili netačan. Neki jezički primjeri bi bili:

- Ja sam gladan. - tačno

- Ja se bojim mraka. - netačno

U formalnoj logici kompjuterskih naukau, mi testiramo odnose među brojevima.

- 15 je veće od 20 - netačno
- 5 je jednako 5 - tačno

Prošli put smo radili varijable, a ovog puta ćemo naučiti kako koristiti varijable u bulovim izrazima i na taj način ćemo omogućiti našem programu da se grana.

- $x > 20$ - ovo ovisi od vrijednosti varijable x
- $y == 5$ - ovisi od varijable y

Relacioni operatori

>	veće od	<=	manje ili jednako od
<	manje od	==	jednakost
>=	veće ili jednako od	!=	nejednakost


Kondicionali: If, Else, Else If

Bulove izrazi, često nazvani i kondicionali, operiraju unutar programa kao pitanja. Da li je 15 veće od 20? Ukoliko je odgovor "da", odnosno "tačno", možemo odabrati da izvršimo određenu instrukciju (na primjer da nacrtamo trokut), ukoliko je odgovor "ne", "netačno", ove se instrukcije mogu ignorisati. Ovo nam uvodi ideju grananja, ovisno od različitih uslova, program može pratiti različite puteve.

Primjer:

Ukoliko je miš na lijevoj strani ekrana, nacrtaj pravougaonik na lijevoj strani.

```
if (mouseX < width/2) {  
  fill(255);  
  rect(0,0,width/2,height);  
}
```



Booleov izraz i rezultirajuće instrukcije u gore navedenom kodu zapravo su formulisane u sljedećoj sintaksi:

```
if (booleov izraz) {  
  //kod koji će se izvršiti ako je booleov izraz tačan
```

```
}
```

Ovaj izraz se može proširiti s ključnom riječi *else* kako bi se uključio kod koji se izvršava ukoliko je booleov izraz netačan:

```
if (booleov izraz) {  
    //kod koji će se izvršiti ako je booleov izraz tačan  
} else {  
    //kod koji će se izvršiti ako je booleov izraz netačan  
}
```

Primjer:

Ukoliko je miš na lijevoj strani ekrana, nacrtaj bijelu pozadinu, u protivnom neka je pozadina crna.


```
if (mouseX < width/2) {  
    background(255);  
} else {  
    background(0);  
}
```



Konačno, za testiranje višestrukih kondicionala, možemo uključiti i *else if*. Kada se koristi *else if*, kondicionalni izrazi se evaluiraju u sljedećem redoslijedu. Prvi booleov izraz koji se pronađe a da je tačan, izvršava se s njim vezani kod, a svi ostali booleovi izrazi će da se ignorišu.

```
if (booleov izraz #1) {  
    //kod koji će se izvršiti ako je booleov izraz #1 tačan  
} else if(booleov izraz #2) {  
    //kod koji će se izvršiti ako je booleov izraz #2 tačan  
} ....  
} else if(booleov izraz #n) {  
    //kod koji će se izvršiti ako je booleov izraz #n tačan  
} else {  
    //kod koji će se izvršiti ako je booleov izraz netačan
```

}



U jednom kondicionalu možete imati samo jedno *if* i samo jedno *else*. No možete imati koliko god želite *else if*.

Primjer:

Ukoliko je miš na lijevoj trećini ekrana, nacrtaj bijelu pozadinu, ukoliko je na srednjoj trećini, nacrtaj sivu, a ukoliko je na desnoj trećini crnu pozadinu.

```
if (mouseX < width/3) {  
background(255);  
}  
else if (mouseX < 2*width/3) {  
background(127);  
}  
else {  
background(0);  
}
```



Zadatak 1:

Sljedeći kod predstavlja primjer u kojem ocjene pretvaramo u slova. Ako je ocjena veća od Ispunite sljedeći kod kako biste upotpunili booloev izraz.

Ocjena A je od 81 - 100

Ocjena B je od 61 - 80

Ocjena C je od 41 - 60

Ocjena D je od 21 - 40

Ocjena E je od 0 - 20

```
float grade = random(0,100);  
  
if (_____) {  
    println( "Assign letter grade A. ");  
} else if (_____) {  
    println (______);  
} else if (_____) {  
    println(______);  
} else if (_____) {  
    println(______);  
} else {  
    println(______);  
}
```

Kondicionali u programu

Primjer:

Korak 1: Kreirajte varijable koje će čuvati vrijednosti za tri boje, crvenu, zelenu i plavu. Dajte im imena: *r*, *g*, *b*.

Korak 2: Konstantno iscrtavajte pozadinu na osnovu tih boja.

Korak 3: Ukoliko je miš na desnoj strani ekrana, povećajte vrijednost varijable *r* za 1, a ukoliko je miš na lijevoj strani ekrana smanjite vrijednost varijable *r* za 1.

Korak 4: Ograničite varijablu *r* da bude unutar opsega 0 do 255.

```
float r = 150;

float g = 0;

float b = 0;

void setup() {
  size(200,200);
}

void draw() {
  background(r,g,b);

  stroke(255);

  line(width/2,0,width/2,height);

  if(mouseX > width/2) {
    r = r + 1;
  } else {
    r = r - 1;
  }

  if (r > 255) {
    r = 255;
  } else if (r < 0) {
    r = 0;
  }
}
```